G01ALF – NAG Fortran Library Routine Document

Note. Before using this routine, please read the Users' Note for your implementation to check the interpretation of bold italicised terms and other implementation-dependent details.

1 Purpose

G01ALF calculates a five-point summary for a single sample.

2 Specification

```
SUBROUTINE G01ALF(N, X, IWRK, RES, IFAIL)
INTEGER           N, IWRK(N), IFAIL
real              X(N), RES(5)
```

3 Description

G01ALF calculates the minimum, lower hinge, median, upper hinge and the maximum of a sample of n observations.

The data consist of a single sample of n observations denoted by x_i and let z_i, for $i = 1, 2, \ldots, n$ represent the sample observations sorted into ascending order.

Let $m = \frac{n}{2}$ if n is even and $\frac{(n+1)}{2}$ if n is odd,

and $k = \frac{m}{2}$ if m is even and $\frac{(m+1)}{2}$ if m is odd.

Then we have:

Minimum	$= z_1$,	
Maximum	$= z_n$,	
Median	$= z_m$	if n is odd,
	$= \frac{z_m + z_{m+1}}{2}$	if n is even,
Lower hinge	$= z_k$	if m is odd,
	$= \frac{z_k + z_{k+1}}{2}$	if m is even,
Upper hinge	$= z_{n-k+1}$	if m is odd,
	$= \frac{z_{n-k} + z_{n-k+1}}{2}$	if m is even.

4 References

[1] Tukey J W (1977) *Exploratory Data Analysis* Addison–Wesley

[2] Erickson B H and Nosanchuk T A (1985) *Understanding Data* Open University Press, Milton Keynes

5 Parameters

1: N — INTEGER *Input*

On entry: number of observations in the sample, n.

Constraint: N ≥ 5.

2: X(N) — ***real*** array *Input*

On entry: the sample observations, $x_1, x_2, \ldots, x_n$.

3: IWRK(N) — INTEGER array *Workspace*

4: RES(5) — ***real*** array *Output*

On exit: RES contains the five-point summary as follows:

RES(1) = the minimum
RES(2) = the lower hinge
RES(3) = the median
RES(4) = the upper hinge
RES(5) = the maximum

5: IFAIL — INTEGER *Input/Output*

On entry: IFAIL must be set to 0, −1 or 1. For users not familiar with this parameter (described in Chapter P01) the recommended value is 0.

On exit: IFAIL = 0 unless the routine detects an error (see Section 6).

6 Error Indicators and Warnings

If on entry IFAIL = 0 or −1, explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors detected by the routine:

IFAIL = 1

On entry, N < 5.

7 Accuracy

The computations are stable.

8 Further Comments

The time taken by the routine is proportional to n.

9 Example

The example program calculates a five-point summary for a sample of 12 observations.

9.1 Program Text

Note. The listing of the example program presented below uses bold italicised terms to denote precision-dependent details. Please read the Users' Note for your implementation to check the interpretation of these terms. As explained in the Essential Introduction to this manual, the results produced may not be identical for all implementations.

```
*     G01ALF Example Program Text
*     Mark 14 Release.  NAG Copyright 1989.
*     .. Parameters ..
      INTEGER          NIN, NOUT
      PARAMETER        (NIN=5,NOUT=6)
      INTEGER          NMAX
      PARAMETER        (NMAX=12)
*     .. Local Scalars ..
      INTEGER          I, IFAIL, N
*     .. Local Arrays ..
      real             RES(5), X(NMAX)
      INTEGER          IWRK(NMAX)
*     .. External Subroutines ..
```

```
      EXTERNAL         G01ALF
*     .. Executable Statements ..
      WRITE (NOUT,*) 'G01ALF Example Program Results'
*     Skip heading in data file
      READ (NIN,*)
      READ (NIN,*) N, (X(I),I=1,N)
      IFAIL = 0
*
      CALL G01ALF(N,X,IWRK,RES,IFAIL)
*
      WRITE (NOUT,*)
      WRITE (NOUT,99999) 'Maximum     ', RES(5)
      WRITE (NOUT,99999) 'Upper Hinge ', RES(4)
      WRITE (NOUT,99999) 'Median      ', RES(3)
      WRITE (NOUT,99999) 'Lower Hinge ', RES(2)
      WRITE (NOUT,99999) 'Minimum     ', RES(1)
      STOP
*
99999 FORMAT (1X,A,F16.4)
      END
```

9.2 Program Data

```
G01ALF Example Program Data
 12
 12.0  9.0  2.0  5.0  6.0  8.0  2.0  7.0  3.0  1.0  11.0  10.0
```

9.3 Program Results

```
 G01ALF Example Program Results

 Maximum              12.0000
 Upper Hinge           9.5000
 Median                6.5000
 Lower Hinge           2.5000
 Minimum               1.0000
```